
' {$STAMP BS2}

' {$PBASIC 2.5}

' --------[Program Description]---------------------------------------------------------

' basic stepper motor control using a 10k POT, and built in 4-phase motor controller on PDB

'

'---------------------------------------------------------------------------------------

' --------[I/O Definitions]-------------------------------------------------------------

PotCW                   PIN       0           ' clockwise POT input

PotCCW                  PIN       1           ' counter-clockwise POT input

Coils                   VAR       OUTB        ' output to stepper coils

' ---------[Constants]------------------------------------------------------------------

Scale                   CON       $100        ' to scale RCTIME

Mitsumi                 CON       48          ' steps / revs by type

Howard                  CON       100         '   "      "    "   "

RevSteps                CON       Mitsumi     ' steps per revolution

NumSteps                CON       4           ' use 4-step sequence

LastStep                CON       NumSteps - 1' last step in sequence

#DEFINE Testing = 0

' ---------[Variables]------------------------------------------------------------------

idx                     VAR       Byte        ' loop counter

stpIdx                  VAR       Nib         ' step pointer

stpDelay                VAR       Byte        ' delay for speed control

rcRt                    VAR       Word        ' rc reading - right

rcLf                    VAR       Word        ' rc reading - left

diff                    VAR       Word        ' difference in readings

' ------------[EEPROM Data]-------------------------------------------------------------

'                                 _____

'                                  ABAB

Step1                   DATA      %1100

Step2                   DATA      %0110

Step3                   DATA      %0011

Step4                   DATA      %1001

' ----------[Initialization]------------------------------------------------------------

Setup:

  DIRB = %1111                                  ' make P4..P7 outputs

  stpDelay = 5                                  ' set step delay

' ----------[Program Code]--------------------------------------------------------------

Demo:

  FOR idx = 1 TO RevSteps                       ' 1 rev forward

    GOSUB Step_Fwd

  NEXT

  PAUSE 200

  FOR idx = 1 TO RevSteps                       ' 1 rev back

    GOSUB Step_Rev

  NEXT

  PAUSE 200

Main:

  HIGH PotCW                                    ' read clockwise position

  PAUSE 1

  RCTIME PotCW, 1, rcRt

  HIGH PotCCW                                   ' read ccw position

  PAUSE 1

  RCTIME PotCCW, 1, rcLf

  rcRt = (rcRt */ Scale) MAX 600                ' set speed limit

  rcLf = (rcLf */ Scale) MAX 600

  diff = ABS(rcRt - rcLf)                       ' get differences

  stpDelay = 100 - (diff / 6) MIN 2             ' calculate step delay

  IF (diff < 25) THEN                           ' allow for dead band

    GOTO Main

  ELSE                                          ' do a step

    IF (rcLf < rcRt) THEN

      GOSUB Step_Fwd

    ELSE

      GOSUB Step_Rev

    ENDIF

  ENDIF

  GOTO Main                                      ' repeat

' ---------[Sub-Routines]--------------------------------------------------------------

' turn stepper clockwise one full step

Step_Fwd:

  stpIdx = stpIdx + 1 // NumSteps                ' point to next step

  GOTO Do_Step

' turn stepper counter-clockwise one full step

Step_Rev:

  stpIdx = stpIdx + LastStep // NumSteps         ' point to previous step

  GOTO Do_Step

' read new step data and output to pins

Do_Step:

  READ (Step1 + stpIdx), Coils                   ' output new coil data

  PAUSE stpDelay                                 ' pause between steps

  RETURN


